

La fonction SI

La fonction SI est une fonction logique qui permet, en fonction d'une expression logique (ou booléenne), de faire un choix entre deux hypothèses. Les formules utilisant la fonction SI sont appelées « formules conditionnelles ». Elles permettent de rendre « intelligentes » vos feuilles de calcul.

Découvrir la fonction

Une formule utilisant la fonction SI (ou formule conditionnelle) se présente de la façon suivante :

=SI(Test; Expression si Test=VRAI; Expression si Test=FAUX)

- **Test** est une expression logique. Une expression logique compte au moins un opérateur logique et deux opérandes.

Opérateur logique	Signification
ET	ET logique
OU	OU logique
=	Égal à
>	Supérieur à
>=	Supérieur ou égal à
<	Inférieur à
<=	Inférieur ou égal à
<>	Différent de

- **Expression si Test=VRAI** est une formule qui peut contenir des fonctions (y compris une autre fonction SI), des calculs, une chaîne de caractères...
- **Expression si Test=FAUX** est une formule qui peut contenir des fonctions (y compris une autre fonction SI), des calculs, une chaîne de caractères...

Afficher un message fixe si une condition est remplie

Supposons que vous utilisiez une feuille de calcul dans laquelle sont stockées des données relatives aux ventes mensuelles de produits. Dans la colonne B se trouvent les ventes de l'année 2004 et dans la colonne C se trouvent les ventes de l'année 2005. Dans la colonne D, il s'agit d'afficher le message En progression si les ventes de 2005 sont supérieures à celles de 2004.

La formule à saisir en D4 est la suivante : **=SI(C4>B4;"En progression";"")** Ensuite, étendez-la, à l'aide de la poignée de recopie, jusqu'à la fin du tableau.

Impact de l'absence d'un argument

Dans l'exemple précédent, si vous écrivez **=SI(C4>B4;"En progression")**, le texte **En progression** aurait bien été affiché dans les cellules pour lesquelles la valeur de la colonne C est supérieure à celle de la colonne B. En revanche, la valeur FAUX aurait été affichée dans les cellules pour lesquelles la valeur de la colonne C est inférieure à celle de la colonne B.

Afficher un message variable si une condition est remplie

Il est possible d'améliorer l'exemple précédent en faisant en sorte que le message affiché indique la valeur de la progression du chiffre d'affaires. Pour cela, il suffit de juxtaposer une chaîne de caractères et une formule de calcul à l'aide de l'opérateur de concaténation (&).

La formule à saisir en D4 est la suivante : **=SI(C4>B4;"En progression de "&(C4-B4)&" ";"")** Ensuite étendez-la, à l'aide de la poignée de recopie, jusqu'à la fin du tableau.

Intégrer le résultat d'une formule conditionnelle dans une expression

Il est possible d'intégrer le résultat d'une fonction dans une autre expression. Ainsi, si le résultat d'une entreprise se trouve en B5 et que vous souhaitiez indiquer « en clair » en B7 qu'il s'agit d'une perte ou d'un bénéfice, la formule est la suivante : **"L'entreprise a réalisé "&SI(B5>0;"un bénéfice de "&B5&" "; "une perte de "&-B5&" ")**

Faire un test sur une chaîne de caractères

Il est possible d'élaborer une formule conditionnelle fondée sur un test mettant en jeu des valeurs numériques, mais aussi des chaînes de caractères.

Pour illustrer cette possibilité, nous allons utiliser une feuille de calcul listant des factures. Dans la colonne A se trouve le nom du fournisseur, dans la colonne B, le numéro de la facture, dans la colonne C, la date d'échéance et enfin, dans la colonne D, le montant. La formule conditionnelle suivante, à saisir en E4, permet de mettre en évidence les factures d'un fournisseur particulier : **=SI(A4="durand";"A surveiller";"OK")** Ensuite, étendez-la, à l'aide de la poignée de recopie, jusqu'à la fin du tableau.

Distinction de casse (majuscules/minuscules)

La condition précédente ne fait pas la distinction entre les minuscules et les majuscules. En effet, les expressions DURAND et Durand donnent un résultat VRAI. Si vous souhaitez faire un test tenant compte de la casse, il faut utiliser la fonction EXACT. Ainsi la formule **=SI(EXACT(A4;"durand");"A surveiller";"OK")** ne renvoie A surveiller que si A4 contient durand, et non Durand ou DURAND.

Il est possible d'utiliser les opérateurs > et < avec des chaînes de caractères. Par exemple "ABC"<"ABD" donne le résultat VRAI. En effet, Excel utilise l'ordre alphabétique pour comparer des chaînes de caractères. Dans ce cas, il n'y a pas de distinction entre les majuscules et les minuscules.

Faire un test sur une date

Les dates servent fréquemment de critères de test. En effet, il n'est pas rare d'avoir à contrôler le dépassement d'un délai, l'occurrence d'une date précise... Pour cela, il est possible d'utiliser une formule conditionnelle fondée sur un test mettant en jeu des dates. Par exemple, si dans la liste de factures, vous souhaitez mettre en évidence celles qui arrivent à échéance avant le 30/11/2009, il faut faire un test sur la date d'échéance afin de déterminer si elle est supérieure ou inférieure au 30/11/2009.

La formule à saisir en E4 est la suivante : **=SI(C4<DATE(2009;11;30);"A surveiller";"OK")** Ensuite, étendez-la, à l'aide de la poignée de recopie, jusqu'à la fin du tableau.

Notez l'emploi de la fonction DATE(année;mois;jour) pour indiquer une date précise. Si vous écriviez C4<30/11/2009, Excel interpréterait l'expression 30/11/2009 comme : 30 divisé par 11, le tout divisé par 2009.

Inclure une formule dans la condition

Afin d'étendre les possibilités des formules conditionnelles, il est possible d'inclure des formules et des fonctions dans les critères de test. Supposons que vous souhaitiez mettre en exergue, parmi une liste de produits, ceux dont le chiffre d'affaires est supérieur à la moyenne. Il faut inclure dans le test la moyenne des chiffres d'affaires, ce qui est possible à l'aide de la fonction MOYENNE.

Si les valeurs à tester se trouvent dans la plage B4:B15, la formule à saisir en C4 est la suivante :

=SI(B4>MOYENNE(\$B\$4:\$B\$15);"Supérieur à la moyenne";"") Ensuite, étendez-la, à l'aide de la poignée de recopie, jusqu'à la fin du tableau.

Utiliser les opérateurs ET et OU

Il faut parfois combiner plusieurs tests logiques pour modéliser une situation « réelle ». Par exemple, supposons que, dans un fichier client, vous souhaitiez faire un traitement particulier pour les hommes de plus de 45 ans dont le salaire est supérieur à 2 000 euros. La formule conditionnelle à créer est fondée sur l'expression suivante : Sexe = "H" ET Âge > 45 ET Salaire > 2 000.

De même, si vous souhaitez mettre en exergue les femmes exerçant le métier de vendeuse ou de comptable, la formule conditionnelle à développer est fondée sur l'expression suivante : Sexe = "F" ET (Profession = "Vendeuse" OU Profession = "Comptable").

Il s'agit en fait de combiner les différentes expressions logiques unitaires (du type A = B, A < B, A > B...), à l'aide de deux opérateurs logiques : ET et OU.

L'opérateur ET

L'opérateur ET renvoie FAUX si l'un des arguments est faux.

A	B	A ET B
FAUX	FAUX	FAUX
FAUX	VRAI	FAUX
VRAI	FAUX	FAUX
VRAI	VRAI	VRAI

L'un des arguments peut être lui-même une formule faisant intervenir des opérateurs logiques ET et OU.

Dans Excel, c'est la fonction ET qui permet de mettre en œuvre l'opérateur logique ET.

Syntaxe : **ET(valeur_logique1;valeur_logique2;...)**
valeur_logique1, valeur_logique2... sont 1 à 255 conditions que vous souhaitez tester et qui peuvent être soit VRAI, soit FAUX.

L'exemple, Sexe = "H" ET Âge > 45 ET Salaire > 2 000, s'écrit :

ET(Sexe="H";Age>45;Salaire>2000)

L'opérateur OU

L'opérateur OU renvoie VRAI si l'un des arguments est vrai.

A	B	A OU B
FAUX	FAUX	FAUX
FAUX	VRAI	VRAI
VRAI	FAUX	VRAI
VRAI	VRAI	VRAI

L'un des arguments peut être lui-même une formule faisant intervenir des opérateurs logiques ET et OU.

Dans Excel, c'est la fonction OU qui permet de mettre en œuvre l'opérateur logique OU.

Syntaxe : **OU(valeur_logique1;valeur_logique2 ;...)**
valeur_logique1, valeur_logique2... sont 1 à 255 conditions que vous souhaitez tester et qui peuvent être soit VRAI, soit FAUX.

L'exemple, Sexe = "F" ET (Profession = "Commercial" OU Profession = "Comptable"), s'écrit :

ET(Sexe="F";OU(Profession="Commercial";Profession="Comptable"))

L'opérateur NON

L'opérateur NON renvoie la valeur inverse de son argument.

A	NON A
FAUX	VRAI
VRAI	FAUX

L'un des arguments peut lui-même être une formule faisant intervenir des opérateurs logiques ET et OU.

Dans Excel, c'est la fonction NON qui permet de mettre en œuvre l'opérateur logique NON.

Syntaxe : **NON(valeur_logique)**
valeur_logique Valeur ou expression qui peut prendre la valeur VRAI ou FAUX.

L'opérateur OU exclusif

L'opérateur logique XOR se comporte de la façon suivante :

A	B	A XOR B
FAUX	FAUX	FAUX
FAUX	VRAI	VRAI
VRAI	FAUX	VRAI
VRAI	VRAI	FAUX

La différence avec l'opérateur OU réside dans la nécessité de n'avoir que l'un ou l'autre (mais pas les deux) des arguments avec une valeur VRAI.

A XOR B s'écrit (A ET NON(B)) OU (NON(A) ET B) soit, selon la syntaxe Excel :

OU(ET(A;NON(B));ET(NON(A);B))

Imbriquer plusieurs fonctions SI

Les formules conditionnelles utilisées jusqu'à présent sont à « un niveau », c'est-à-dire que le résultat est fonction d'un seul test, aussi compliqué soit-il. Cependant, certaines problématiques nécessitent l'élaboration de véritables arbres de décisions, du type :

Si **condition1** alors

 Si **condition2** alors

 Si **condition3** alors

 Action1

```

Sinon
    Action2
Sinon
    Si condition4 alors
        Action3
    Sinon
        Action4
Sinon
    Si condition5 alors
        Si condition6 alors
            Action5
        Sinon
            Action6
Sinon
    Si condition7 alors
        Action7
    Sinon
        Action8

```

Rendre les formules plus lisibles

Pour rendre vos formules complexes plus lisibles, il est possible d'insérer des sauts de ligne lors de la saisie, à l'aide de la combinaison de touches **Alt** + **Entrée**. La formule précédente peut ainsi s'écrire :

```

=SI(condition1;
SI(condition2;
SI(condition3;Action1;Action2);SI(condition4;Action3;Action4));
SI(condition5;
SI(condition6;Action5;Action6);SI(condition7;Action7;Action8)))

```

Découvrir la fonction CHOISIR

La fonction CHOISIR peut constituer, dans certains cas précis, une solution plus élégante que l'imbrication de fonctions SI. En effet, la fonction CHOISIR permet de sélectionner une valeur dans une liste, en fonction d'un numéro.

CHOISIR Utilise l'argument `no_index` pour renvoyer une des valeurs de la liste des arguments de valeurs.

Syntaxe : **CHOISIR(no_index;valeur1;valeur2;...)**

no_index Spécifie quel argument de valeur doit être sélectionné. L'argument `no_index` doit être un nombre compris entre 1 et 29, une formule ou une référence à une cellule contenant un nombre compris entre 1 et 29.

valeur1, valeur2... De 1 à 255 arguments de valeurs parmi lesquels CHOISIR sélectionne une valeur ou une action à exécuter en fonction de l'argument `no_index` spécifié. Ces arguments peuvent être des nombres, des références de cellules, des noms définis, des formules, des fonctions ou du texte.

Pour illustrer l'emploi de cette fonction, supposons que vous souhaitiez afficher en toutes lettres le jour de la semaine à partir de son numéro (1 = Lundi, 2 = Mardi...). Si le numéro du jour se trouve en B3, vous pouvez écrire en C3 :

```
=CHOISIR(B3;"Lundi";"Mardi";"Mercredi";"Jeudi";"Vendredi";"Samedi";"Dimanche")
```

Si vous saisissez un numéro inférieur à 1 ou supérieur à 7, le message d'erreur #VALEUR! apparaît en C3.

Il est possible d'utiliser des références à des plages de cellules en guise d'arguments. Ainsi, supposons que vous disposiez des chiffres d'affaires mensuels d'une entreprise dans la plage B2:B13 et que vous souhaitiez afficher (en D5) les chiffres d'affaires trimestriels à la demande (en fonction du numéro du trimestre se trouvant en D2). La formule en D5 est la suivante : **=SOMME(CHOISIR(D2;B2:B4;B5:B7;B8:B10;B11:B13))**

Dans ce cas, CHOISIR renvoie une plage, qui est à son tour un argument pour la fonction SOMME.

Les autres fonctions d'Excel.

Les fonctions essentielles d'Excel se divisent en trois catégories :

- Les fonctions de calcul ;
- Les fonctions d'arrondi ;
- Les fonctions de comptage.

Fonctions de calcul

Il s'agit de fonctions permettant de faire des calculs simples (somme, moyenne), mais également de réaliser des traitements statistiques plus complexes.

MAX

Renvoie le plus grand nombre de la série de valeurs.

Syntaxe : **MAX(nombre1;nombre2;...)**

nombre1;nombre2... 1 à 255 nombres parmi lesquels vous souhaitez trouver la valeur la plus grande.

MIN

Renvoie le plus petit nombre de la série de valeurs.

Syntaxe : **MIN(nombre1;nombre2;...)**

nombre1;nombre2... 1 à 255 nombres parmi lesquels vous souhaitez trouver la valeur minimale.

MOD

Renvoie le reste de la division de l'argument nombre par l'argument diviseur. Le résultat est du même signe que diviseur.

Syntaxe : **MOD(nombre;diviseur)**

nombre Nombre à diviser pour obtenir le reste.

diviseur Nombre par lequel vous souhaitez diviser nombre.

MODE

Renvoie la valeur la plus fréquente ou la plus répétitive dans une matrice ou une plage de données.

Syntaxe : **MODE(nombre1;nombre2;...)**

nombre1;nombre2... 1 à 255 arguments dont vous souhaitez déterminer le mode. Vous pouvez également utiliser une matrice unique ou une référence à une matrice, au lieu d'arguments séparés par des points-virgules.

MOYENNE

Renvoie la moyenne (arithmétique) des arguments.

Syntaxe : **MOYENNE(nombre1;nombre2;...)**

nombre1;nombre2... 1 à 255 arguments numériques dont vous voulez obtenir la moyenne.

QUOTIENT

Renvoie la partie entière du résultat d'une division. Utilisez cette fonction lorsque vous voulez ignorer le reste d'une division.

Syntaxe : **QUOTIENT(numérateur;dénominateur)**

numérateur Le nombre à diviser.

dénominateur Diviseur.

RANG

Renvoie le rang d'un nombre dans une liste d'arguments.

Syntaxe : **RANG(nombre;référence;ordre)**

nombre Nombre dont vous voulez connaître le rang.

référence Matrice ou référence à une liste de nombres. Les valeurs non numériques dans référence sont ignorées.

ordre Numéro qui spécifie comment déterminer le rang de l'argument nombre (0 ou omis : ordre croissant ; 1 : ordre décroissant).

SOMME

Additionne tous les nombres contenus dans une plage de cellules.

Syntaxe : **SOMME(nombre1;nombre2;...)**

nombre1;nombre2... 1 à 255 arguments dont vous voulez calculer la somme.

SOMME.SI

Additionne des cellules spécifiées si elles répondent à un critère donné.

Syntaxe : **SOMME.SI(plage;critère;somme_plage)**

plage Plage de cellules sur lesquelles s'applique le critère.

Critère Critère, sous forme de nombre, d'expression ou de texte, définissant les cellules à additionner.

somme_plage Cellules à additionner.

Fonctions de comptage

Ces fonctions sont indispensables si vous devez résoudre des problèmes de dénombrement de cellules respectant des critères particuliers.

COLONNES

Renvoie le nombre de colonnes d'une matrice ou d'une référence.

Syntaxe : **COLONNES(tableau)**

tableau Formule matricielle, référence d'une plage de cellules ou tableau dans lequel vous souhaitez compter le nombre de colonnes.

LIGNES

Renvoie le nombre de lignes d'une matrice ou d'une référence.

Syntaxe : **LIGNES(tableau)**

tableau Formule matricielle, référence d'une plage de cellules ou tableau dans lequel vous voulez compter le nombre de lignes.

NB

Détermine le nombre de cellules contenant des nombres et les nombres compris dans la liste des arguments.

Syntaxe : **NB(valeur1;valeur2;...)**

valeur1;valeur2... 1 à 255 arguments qui peuvent contenir ou faire référence à différents types de données, mais seuls les nombres sont comptés.

NB.SI

Compte le nombre de cellules à l'intérieur d'une plage qui répondent à un critère donné.

Syntaxe : **NB.SI(plage;critère)**

plage Plage de cellules dans laquelle vous voulez compter les cellules.

critère Critère, exprimé sous forme de nombre, d'expression ou de texte, qui détermine les cellules à compter.

NB.VIDE

Compte le nombre de cellules vides à l'intérieur d'une plage de cellules spécifiée.

Syntaxe :	NB.VIDE(plage)
plage	Plage dans laquelle vous voulez compter les cellules vides.

Faire des calculs

Faire des calculs, des plus simples aux plus ardu, est la vocation première d'Excel. Nous allons aborder des problématiques classiques auxquelles vous avez sûrement été confronté : calculer des sommes, des moyennes, faire des calculs en fonction de critères...

Calculer une somme

Supposons que vous disposiez d'une feuille de calcul destinée à enregistrer des factures relatives à un projet. Pour connaître le total des montants des factures saisies (les montants sont dans la plage E4:E13), saisissez dans n'importe quelle cellule, la formule suivante : **=SOMME(E4:E13)**

Calculer un cumul glissant

Le cumul glissant dans une colonne permet d'obtenir le total des valeurs comprises entre la ligne courante et la première ligne. Cette approche permet de matérialiser la progression des valeurs. Si les valeurs se trouvent dans la colonne E (à partir de la cellule E4), saisissez la formule suivante en F4 : **=E4** Saisissez en F5 : **=F4+E5**
Étendez ensuite le contenu jusqu'à la dernière ligne.

L'inconvénient de cette approche réside dans l'affichage du cumul même s'il n'y a pas de valeur dans la colonne E. Pour remédier à cela, modifiez la formule se trouvant en F5 : **=SI(E5<>"";F4+E5;"")**
Étendez ensuite la formule à l'ensemble de la colonne.

Dès que vous saisissez une nouvelle valeur, le cumul apparaîtra dans la ligne correspondante.

Calculer une moyenne

Pour connaître la moyenne des montants des factures saisies (les montants sont dans la plage E4:E13), saisissez dans n'importe quelle cellule, la formule suivante : **=MOYENNE(E4:E13)**

Calculer la somme de cellules respectant des critères

Dans certaines situations, la fonction SOMME n'est pas suffisamment « fine » pour résoudre un problème. Dans notre exemple de liste de factures, supposons que vous souhaitiez totaliser les montants supérieurs à 300 euros. Pour ce faire, vous avez besoin de la fonction SOMME.SI, qui permet de calculer une somme de valeurs respectant certains critères. La formule est la suivante : **=SOMME.SI(E4:E13;">300")**

En toute rigueur, la fonction SOMME.SI a besoin de trois arguments :

- la plage de cellules sur laquelle doit porter le test ;
- Le test ;
- La plage de cellules à totaliser.

Dans ce cas, la plage de cellules à totaliser a été omise car elle est identique à la plage à tester.

À présent, supposons que vous souhaitiez totaliser les montants des factures du fournisseur « ABC ».

La formule est alors la suivante : **=SOMME.SI(A4:A13;"ABC";E4:E13)**

Dans ce cas, la plage à tester (les noms des fournisseurs) est différente de la plage à totaliser. Vous pouvez inclure des caractères génériques dans les tests. Ainsi la formule suivante totalise les montants des factures des fournisseurs dont le nom contient la chaîne de caractères « ur », quel que soit son emplacement : **=SOMME.SI(A4:A13;"*ur*";E4:E13)**

Le caractère * remplace une chaîne de caractères, quelle que soit sa taille. Le caractère ? Remplace un caractère unique. La formule suivante totalise les montants des factures dont le numéro débute par F13 : **=SOMME.SI(B4:B13;"F13?";E4:E13)**

Enfin, si vous souhaitez utiliser des seuils facilement paramétrables dans vos sommes conditionnelles, vous pouvez faire référence à des cellules dans les tests. Ainsi la formule suivante totalise les factures dont le montant est inférieur ou égal au contenu de la cellule H3 : **=SOMME.SI(E4:E13;"<="&H3)**

Vous pouvez aussi utiliser des formules dans les critères : **=SOMME.SI(E4:E13;">"&MOYENNE(E4:E13))**

La formule précédente totalise les montants supérieurs à moyenne.

Combiner les critères

Pour totaliser des valeurs comprises entre deux bornes, il faut faire appel deux fois à la fonction SOMME.SI. Par exemple, pour totaliser les valeurs comprises entre 100 et 200, il faut retrancher le total des valeurs supérieures ou égales à 200 de celui des valeurs supérieures à 100: **=SOMME.SI(E4:E13;">100")-SOMME.SI(E4:E13;">=200")** Cette méthode permet de simuler un opérateur logique ET. Pour obtenir un OU, ajoutez les sommes conditionnelles correspondant à chaque critère.

Déterminer la valeur la plus fréquente dans une plage de cellules

Dans le cas de réponses à un questionnaire ou d'un vote à la majorité relative, il est utile de déterminer la valeur la plus fréquente afin de connaître l'avis dominant. Si les valeurs se trouvent dans la plage B3:B22, la valeur la plus fréquente est obtenue à l'aide de la formule suivante : **=MODE(B3:B22)**

La fonction MODE ne fonctionne qu'avec des valeurs numériques. Si vos données sont des chaînes de caractères, il faut leur affecter une codification numérique, sur le principe du questionnaire à choix multiple.

Dénombrer des cellules

Afin d'analyser le contenu d'une feuille de calcul, il est souvent intéressant de compter les cellules qui respectent certains critères : cellules vides, non vides, contenant des valeurs numériques, du texte... Excel dispose de plusieurs fonctions qui vous seront utiles pour mener à bien cette tâche.

Compter les cellules vides

Pour compter les cellules vides dans une plage de cellules (ici B4:C15), saisissez la formule suivante :

=NB.VIDE(B4:C15)

Pour comptabiliser les cellules vides dans une colonne (la B, par exemple), saisissez la formule suivante :

=NB.VIDE(B:B)

Pour un comptage au niveau des lignes (1 à 3 par exemple), saisissez la formule suivante :

=NB.VIDE(1:3)

Compter les cellules contenant des valeurs numériques

Pour compter les cellules contenant des valeurs numériques dans une plage de cellules (ici B4:C15), saisissez la formule suivante : **=NB(B4:C15)**

Pour comptabiliser les cellules contenant des valeurs numériques dans une colonne (la B, par exemple), saisissez la formule suivante : **=NB(B:B)**

Pour un comptage au niveau des lignes (1 à 3, par exemple), saisissez la formule suivante : **=NB(1:3)**

Les dates sont considérées comme des valeurs numériques.

Compter les cellules contenant du texte

Pour compter les cellules contenant du texte dans une plage de cellules (ici B4:C15), saisissez la formule suivante : **=NBVAL(B4:C15)-NB(B4:C15)**

Il s'agit en fait de retrancher au nombre total de cellules le nombre de cellules contenant une valeur numérique. Cette formule ne donne pas le bon résultat si des messages d'erreur sont présents dans la plage : ces derniers sont comptabilisés comme du texte.

Compter les cellules contenant une chaîne de caractères

Pour compter les cellules contenant une chaîne de caractères précise (par exemple test) dans une plage de cellules (ici B4:C15), saisissez la formule suivante : **=NB.SI(B4:C15;"test")**

Vous pouvez utiliser des caractères génériques pour remplacer un ou plusieurs caractères. La formule suivante comptabilise toutes les cellules qui contiennent l'expression fact dans la colonne B.

=NB.SI(B:B;"*fact*")

La formule suivante comptabilise toutes les cellules de la colonne B qui contiennent une chaîne de quatre caractères : **=NB.SI(B:B;"????")**

Compter les cellules dont le contenu est supérieur à un seuil

Pour compter les cellules qui contiennent des valeurs supérieures à un seuil, vous pouvez utiliser la fonction NB.SI. Ainsi, la formule suivante permet de compter le nombre de cellules dont le contenu est supérieur à 200 :

=NB.SI(B4:C15;">200")

Si vous souhaitez fonder le critère sur le contenu d'une cellule (par exemple F2), et non sur une valeur fixe, la formule à employer est la suivante : **=NB.SI(B4:C15;">"&F2)**

Similitudes avec SOMME.SI

Tout ce qui a été dit dans ce chapitre concernant la fonction SOMME.SI est applicable à la fonction NB.SI.

Comprendre la représentation des dates dans Excel

Avant d'entrer dans le vif du sujet, il convient au préalable de bien expliciter les principes qui fondent la représentation des dates dans Excel.

Découvrir la notion de numéro de série

Excel enregistre les dates sous la forme de nombres séquentiels appelés « numéros de série ». Par défaut, le 1er janvier 1900 correspond au numéro de série 1, et le 1er janvier 2010 au numéro de série 40179 car 40 179 jours se sont écoulés depuis le 1er janvier 1900. Excel enregistre les heures sous la forme de fractions décimales car

l'heure est considérée comme une partie de la journée. C'est pourquoi 0,5 correspond à l'heure de midi, et 0,75 à 18 heures.

Les dates et les heures étant des valeurs, elles peuvent être ajoutées, soustraites et incluses dans d'autres calculs. Pour afficher une date sous la forme d'un numéro de série et une heure sous la forme d'une fraction décimale, affectez le format Standard à la cellule contenant la date ou l'heure.

Saisir des dates et des heures

Si vous saisissez 04/12/09, Excel affiche 04/12/2009. Il s'agit uniquement d'un format ; il n'a pas d'impact sur la valeur. En d'autres termes, Excel reconnaît l'entrée 04/12/09 comme une date valide, la convertit en numéro de série et formate le résultat sous la forme jj/mm/aaaa. Le contenu de la cellule est bien une valeur numérique (numéro de série), associée à un format de date.

Pour saisir une date dans Excel, séparez les jours, mois et années par des / ou des -. Si vous saisissez le mois en lettres, vous pouvez séparer le jour, le mois et l'année par des espaces. Voici quelques exemples de saisies correctes :

- 04/12/09 est affiché 04/12/2009.
- 04-12-09 est affiché 04/12/2009.
- 04 décembre 2009 est affiché 04-déc-2009.
- 04/12 est affiché 04-déc.

Lors d'une saisie de date, si le jour n'est pas indiqué, la date correspond au premier jour du mois (ainsi 12/2009 est affiché déc-2009 mais la cellule contient la valeur 01/12/2009).

Pour la saisie des heures, seul le caractère : est autorisé pour séparer les heures et les minutes. Voici quelques exemples de saisies correctes :

- 23:6 est affiché 23:06.
- 23: est affiché 23:00.
- 45:12 est affiché 45:12:00.

Les fonctions indispensables des dates

ANNEE

Renvoie l'année correspondant à une date. L'année est renvoyée sous la forme d'un nombre entier dans la plage 1900-9999.

Syntaxe :	ANNEE(numéro_de_série)
numéro_de_série	Numéro de série de la date dont vous voulez trouver l'année.

	A	B	C	D	E
1					
2	24/04/2005	2005	←	=ANNEE(A2)	
3		2001	←	=ANNEE("24/06/2001")	
4		2010	←	=ANNEE(AUJOURDHUI())	

AUJOURDHUI

Renvoie le numéro de série de la date courante.

Syntaxe :	AUJOURDHUI()
-----------	--------------

	A	B	C	D
6				
7	Aujourd'hui :	27/02/2010	←	=AUJOURDHUI()
8	Demain :	28/02/2010	←	=AUJOURDHUI()+1
9	Hier :	26/02/2010	←	=AUJOURDHUI()-1
10				

DATE

Renvoie le numéro de série séquentiel qui représente une date particulière.

Syntaxe :	DATE(année,mois,jour)
année	Argument pouvant compter entre un et quatre chiffres. Excel interprète l'argument année en fonction du système de dates que vous utilisez.
mois	Nombre représentant le mois de l'année.
jour	Nombre représentant le jour du mois.

	A	B	C	D	E	F
12						
13	Commande:	Année:	2010			
14		Mois:	4			
15		Jour:	12			
16						
17	Délai de livraison:		15			
18						
19	Date de livraison :		27/04/2010			=DATE(C14;C15;C16)+C18
20						

DATEDIF

Calcule la différence entre deux dates en années, mois et jours.

Syntaxe :	DATEDIF(date_début,date_fin;unité)
date_début	Date de début.
date_fin	Date de fin.
unité	Indique en quelle unité doit être calculée la différence entre les deux dates.

L'argument unité peut prendre les valeurs présentes dans le tableau suivant :

Valeur de l'argument	Signification
"y"	Différence en années
"m"	Différence en mois
"d"	Différence en jours
"ym"	Différence en mois, une fois les années soustraites
"yd"	Différence en jours, une fois les années soustraites
"md"	Différence en jours, une fois les années et les mois soustraits

	A	B	C	D	E	F	G
1	Date d'entrée :	23/10/1995					
2	Date du jour :	27/02/2010					
3							
4	Ancienneté:	14	ans	4	mois	4	jours
5		↑		↑		↑	
6		=DATEDIF(B1;B2;"y")				=DATEDIF(B1;B2;"md")	
7							
8				=DATEDIF(B1;B2;"ym")			
9							

FIN.MOIS

Renvoie le numéro de série du dernier jour du mois précédant ou suivant date_départ du nombre de mois indiqué.

Syntaxe :	FIN.MOIS(date_départ;mois)
date_départ	Date de début.
mois	Nombre de mois avant ou après date_départ. Une valeur de mois positive donne une date future, tandis qu'une valeur négative donne une date passée.

	A	B	C	D	E	F	G
28							
29			Date facture:	25/02/2010			
30							
31		Règlement 30 jours nets:		27/03/2010	←	=+D29+30	
32		Règlement 30 jours fin de mois :		31/03/2010	←	=FIN.MOIS(D29+30;0)	
33		Règlement 30 jours fin de mois le 10 :		10/04/2010	←	=FIN.MOIS(D29+30;0)+10	
34							

HEURE

Renvoie l'heure correspondant à la valeur de l'heure fournie. L'heure est un nombre entier compris entre 0 (12:00 AM) et 23 (11:00 PM).

Syntaxe :	HEURE(numéro_de_série)
numéro_de_série	Code de temps correspondant à l'heure que vous voulez trouver. Les codes de temps peuvent être entrés sous la forme de chaînes de caractères entre guillemets (par exemple, "6:45 PM"), de caractères décimaux (par exemple, 0,78125, qui représente 6:45 PM) ou de résultats d'autres formules ou fonctions (TEMPSVAL("6:45 PM")).

	A	B	C	D	E
44					
45			17	←	=HEURE(MAINTENANT())
46					
47	24/04/2005 09:45		9	←	=HEURE(A47)
48					

JOUR

Renvoie le jour du mois correspondant à l'argument numéro_de_série. Ce jour est représenté sous la forme d'un nombre entier compris entre 1 et 31.

Syntaxe :	JOUR(numéro_de_série)
numéro_de_série	Code de date du jour que vous voulez trouver.

	A	B	C	D	E
49					
50			27	←	=JOUR(MAINTENANT())
51					
52	24/04/2010 09:45		24	←	=JOUR(A52)
53					

JOURSEM

Renvoie le jour de la semaine correspondant à une date. Par défaut, le jour est donné sous forme d'un nombre entier compris entre 0 et 7.

Syntaxe :	JOURSEM(numéro_de_série;type_retour)
numéro_de_série	Numéro séquentiel représentant la date du jour que vous cherchez.
type_retour	Chiffre qui détermine le type d'information que la fonction renvoie.

Voici la liste des valeurs possibles du type_retour

Valeur de type_retour	Chiffre renvoyé
1 ou omis	Chiffre compris entre 1 (dimanche) et 7 (samedi)
2	Chiffre compris entre 1 (lundi) et 7 (dimanche)
3	Chiffre compris entre 0 (lundi) et 6 (dimanche)

	A	B	C	D	E	F
61						
62		Date de naissance:	16/10/1971			
63						
64	Vous êtes né un	Samedi				
65		↑				
66						
67	=CHOISIR(JOURSEM(C62);"Dimanche";"Lundi";"Mardi";"Mercredi";"Jeudi";"Vendredi";"Samedi")					

MAINTENANT

Donne le numéro de série de la date et de l'heure courantes.

Syntaxe : MAINTENANT()

MOIS

Renvoie le mois d'une date représentée par l'argument numéro_de_série. Le mois est donné sous la forme d'un nombre entier compris entre 1 (janvier) et 12 (décembre).

Syntaxe :	MOIS(numéro_de_série)
numéro_de_série	Code de date du mois que vous voulez trouver.

MOIS.DECALER

Renvoie le numéro de série qui représente la date correspondant à une date spécifiée (l'argument date_départ), corrigée du nombre de mois indiqué.

Syntaxe :	MOIS.DECALER(date_départ;mois)
date_départ	Date qui définit la date à partir de laquelle doit s'appliquer le décalage.
mois	Nombre de mois avant ou après date_départ. Une valeur de mois positive donne une date future, tandis qu'une valeur négative donne une date passée.

	A	B	C	D	E	F
82						
83		Date du jour :	25/04/2010			
84		Un mois plus tard :	25/05/2010	←	=MOIS.DECALER(C83;1)	
85		Deux mois plus tard :	25/06/2010	←	=MOIS.DECALER(C83;2)	
86		Un mois plus tôt :	25/03/2010	←	=MOIS.DECALER(C83;-1)	
87						

NO.SEMAIN

Renvoie le numéro d'ordre de la semaine dans l'année.

Syntaxe :	NO.SEMAIN(numéro_de_série;méthode)
numéro_de_série	Date de la semaine.
méthode	Détermine quel jour est considéré comme le début de la semaine. La valeur par défaut est 1.

	A	B	C	D	E
107					
108	01/01/2010	1	←	=NO.SEMAIN(A108)	
109	25/01/2010	5	←	=NO.SEMAIN(A109)	
110	12/03/2010	11	←	=NO.SEMAIN(A110)	
111	31/08/2010	36	←	=NO.SEMAIN(A111)	
112	30/12/2010	53	←	=NO.SEMAIN(A112)	